Normalization by evaluation for λ^{→2}

Thorsten Altenkirch^{1 ⋆} and Tarmo Uustalu^{2 ⋆⋆}

1 School of Computer Science and IT, University of Nottingham Nottingham NG8 1BB, United Kingdom

txa@cs.nott.ac.uk

2 Institute of Cybernetics, Tallinn Technical University Akadeemia tee 21, EE-12618 Tallinn, Estonia

tarmo@cs.ioc.ee

Abstract. We show that the set-theoretic semantics for λ^{→2} is complete by inverting evaluation using decision trees. This leads to an implementation of normalization by evaluation which is witnessed by the source of this paper being a literate Haskell script. We show the correctness of our implementation using logical relations.

1 Introduction

Which is the simplest typed λ-calculus without uninterpreted base types or type variables? We suggest that the answer should be λ^{→2}: simply typed lambda calculus extended by the type of booleans Bool with True, False : Bool and

If t u_0 u_1 : σ, given t : Bool and u_0, u_1 : σ. The equational theory is given by

the usual βη-equations of λ^→ and the following equations concerning Bool:

  If True u_0 u_1 =_{βη} u_0
 If False u_0 u_1 =_{βη} u_1
If t True False =_{βη} t

v (If t u_0 u_1) =_{βη} If t (v u_0) (v u_1)

The equations are motivated by the categorical interpretation of Bool as a boolean object, i.e., an object Bool such that Hom(Γ × Bool, A) ≃ Hom(Γ, A) × Hom(Γ, A) (naturally in Γ and A). The calculus can thus be interpreted in any cartesian closed category with Bool (using the cartesian structure to interpret contexts).

The equational theory introduces some interesting equalities. E.g., consider

once = λf : Bool → Bool. λx : Bool. f x thrice = λf : Bool → Bool. λx : Bool.f (f (f x))

⋆ partially supported by EU Framework 5 thematic networks TYPES and APPSEM II and the Estonian IT Foundation ⋆⋆ partially supported by the Estonian Science Foundation under grant No. 5567

We observe that once =_{βη} thrice. To see this, we note that, given f : Bool →

Bool, we have

f (f (f True)) =_{βη} If (f True) (f (f True)) (f (f False)) =_{βη} If (f True) True (f (f False)) =_{βη} If (f True) True (If (f False) (f True) (f False)) =_{βη} If (f True) True (If (f False) False False) =_{βη} If (f True) True False =_{βη} f True

Symmetrically, we can show that f (f (f False)) =_{βη} f False, and hence

thrice = λf : Bool → Bool. λx : Bool. f (f (f x)) =_{βη} λf : Bool → Bool. λx : Bool. If x (f (f (f True))) (f (f (f False))) =_{βη} λf : Bool → Bool.λx : Bool. If x (f True) (f False) =_{βη} λf : Bool → Bool. λx : Bool.f x = once

It is easy to see that once and thrice are equal in the standard semantics where

Bool is interpreted by a two-element set Bool = {true, false} and function types

are set-theoretic function spaces. We observe that there are only four elements in Bool → Bool = {x → true, x → x, x → ¬x, x → false} and that for all the four f ∈ Bool → Bool we have f^3 = f.

May we use set-theoretic reasoning to prove equalities up to βη-convertibility? The answer is yes for λ^{→2}, because for λ^{→2} we can invert set-theoretic evaluation of typed closed terms. That is: we can define a function quote^σ ∈ �σ�^{Set} → Tm σ such that t =_{βη} quote^σ �t�^{Set}, for any t ∈ Tm σ. Consequently, we get that, for any t, t^′ ∈ Tm σ, t =_{βη} t^′ ⇐⇒ �t�^{Set} = �t^′�^{Set}.

The construction can be carried out in a constructive setting, e.g., using Martin-Löf’s Type Theory as the metalanguage, and we obtain an implementation of normalization nf^σ t = quote^σ �t�^{Set}. We shall here use the language Haskell as a poor man’s Type Theory and obtain a Haskell program to normalize terms. Our implementation is based on decision trees which lead to reasonably simple normal forms.

The existence of quote also implies that =_{βη} is maximally consistent, i.e., identifying any two non-βη-convertible closed terms would lead to an inconsistent theory. This provides another justification for the specific choice of =_{βη}.

Related work

Inverting evaluation to achieve normalization by evaluation (NBE, aka. reductionfree normalization) was pioneered in [5] for simply typed lambda calculus with type variables and a non-standard semantics; a categorical account in terms of presheaves was given in [2]; this was extended to System F in [3, 4]; see [8] for a recent survey on NBE. The completeness of the set-theoretic model in the presence of coproducts has been shown in [7] and our case arises as a special case when there are no type variables. Normalization procedures for typed λcalculus with coproducts can be found in [9, 10] using rewriting techniques and [1] using NBE and sheaf theory. Both approaches allow type variables but do not handle the empty type. Here we present a much simpler construction for closed types using the simplest possible semantics of first-order simply typed lambda calculi—the set-theoretic one—and also provide a concrete implementation of quote and nf, whose correctness we show in detail.

2 Implementation of λ^{→2}

The source code of this paper is a literate Haskell script implementing normalization for λ^{→2} and is available from

http://www.cs.nott.ac.uk/~txa/Nbe2.lhs

We start by introducing types Ty ∈ ⋆, typing contexts Con ∈ ⋆ and untyped terms UTm ∈ ⋆ of the object language by the following Haskell datatype definitions:

data Ty = Bool | Ty :-> Ty
        deriving (Show, Eq)

type Con = [ (String, Ty) ]

data UTm = Var String
        | TTrue | TFalse | If UTm UTm UTm
        | Lam Ty String UTm | App UTm UTm
        deriving (Show, Eq)

We view these recursive definitions as inductive definitions, i.e., we do not consider infinite terms. All the functions we define are total on finite terms.

Implementing typed terms Tm ∈ Con → Ty → ⋆ would take inductive families, which we cannot use in Haskell. But we can implement type inference infer ∈ Con → UTm → Maybe Ty (where Maybe X = 1 + X as usual):

infer :: Con -> UTm -> Maybe Ty
infer gamma (Var x) =
    do sigma <- lookup x gamma
       Just sigma
infer gamma TTrue = Just Bool
infer gamma TFalse = Just Bool
infer gamma (If t u0 u1) =
    do Bool <- infer gamma t
       sigma0 <- infer gamma u0
       sigma1 <- infer gamma u1
       if sigma0 == sigma1 then Just sigma0 else Nothing
infer gamma (Lam sigma x t) =
    do tau <- infer ((x, sigma) : gamma) t
       Just (sigma :-> tau)
infer gamma (App t u) =
    do (sigma :-> tau) <- infer gamma t
       sigma’ <- infer gamma u
       if sigma == sigma’ then Just tau else Nothing

This implementation is correct in the sense that t ∈ Tm_Γ σ iff infer_Γ t = Just σ.

Evaluation of types �−� : Ty → ⋆ in a semantics is again an inductive family, which we cannot do in Haskell, and the workaround is to have all �σ� coalesced into one metalanguage type U (U for ‘untyped elements’) much the same way as all Tm_Γ σ appear coalesced in UTm. We use a type class Sem to state what we require of such a coalesced type U:

class Sem u where
    true :: u
    false :: u
    xif :: u -> u -> u -> u
    lam :: Ty -> (u -> u) -> u
    app :: u -> u -> u

We introduce untyped environments UEnv_U ∈ ⋆:

type UEnv u = [ (String, u) ]

And now we are ready to implement evaluation of terms �−� ∈ UEnv_U → UTm → U:

eval :: Sem u => UEnv u -> UTm -> u
eval rho (Var x) = d
    where (Just d) = lookup x rho
eval rho TTrue = true
eval rho TFalse = false
eval rho (If t u0 u1) = xif (eval rho t) (eval rho u0) (eval rho u1)
eval rho (Lam sigma x t) = lam sigma (\ d -> eval ((x, d) : rho) t)
eval rho (App t u) = app (eval rho t) (eval rho u)

We notice that �−� is type-sound: if ρ ∈ Env_Γ and t ∈ Tm_Γ σ, then �t�_ρ ∈ �σ�.

The standard set-theoretic semantics can be represented in Haskell as an instance of Sem:

data SU = STrue | SFalse | SLam Ty (SU -> SU)

instance Sem SU where
    true = STrue
    false = SFalse
    xif STrue d _ = d
    xif SFalse _ d = d
lam = SLam
app (SLam _ f) d = f d

We write �−�^{Set} to emphasize evaluation in the set-theoretic semantics (this is automatically inferred by Haskell).

Since sets form a cartesian closed category with a boolean object, the settheoretic semantics validates all βη-equalities. This is to say that �−�^{Set} is equationally sound:

Proposition 2.1 (Soundness). If ρ ∈ Env_Γ and t =_{βη} t^′ ∈ Tm_Γ σ, then �t�^{Set}_ρ = �t^′�^{Set}_ρ .

Since all the sets we consider are finite, semantic equality can be implemented in Haskell, by making use of the function enum ∈ (σ ∈ Ty) → Tree �σ�, which we will provide later:

instance Eq SU where
    STrue == STrue = True
    SFalse == SFalse = True
    (SLam sigma f) == (SLam _ f’) =
        and [f d == f’ d | d <- flatten (enum sigma)]
    _ == _ = False

Using on the same function we can also print elements of SU:

instance Show SU where
    show STrue = "STrue"
    show SFalse = "SFalse"
    show (SLam sigma f) =
        "SLam " ++ (show sigma) ++ " " ++
        (show [ (d, f d) | d <- flatten (enum sigma) ])

The equational theory of the calculus itself gives rise to another semantics— the free semantics, or typed closed terms up to βη-convertibility. This semantics can be implemented as another instance of Sem (although of course we have to replace typed closed terms with UTm). To do this properly, we would have to take care of fresh name generation. But it turns out that in our uses of typed closed terms later as names of semantic elements there will be at most one bound variable at any time. We also notice that we get shorter normal forms by implementing a redundancy-avoiding version of If which produces a shorter but βη-equivalent term:

instance Sem UTm where
    true = TTrue
    false = TFalse
    xif t TTrue TFalse = t
    xif t u0 u1 = if u0 == u1 then u0 else If t u0 u1
    lam sigma f = Lam sigma "x" (f (Var "x"))
    app = App

Also this semantics is clearly equationally sound.

3 Implementation of quote

We now proceed to implementing quote ∈ (σ ∈ Ty) → �σ�^{Set} → Tm σ.

To define quote^{σ→τ} we use enum^σ , which generates a decision tree whose leaves are all the elements of �σ�, and questions^σ , which generates the list of questions, i.e. elements of �σ� → �Bool�, based on answers to whom an element of �σ� can be looked up in the tree enum^σ .

Decision trees Tree ∈ Ty → ⋆ are provided by

data Tree u = Val u | Choice (Tree u) (Tree u) deriving (Show, Eq)

We will exploit the fact that Tree is a monad

instance Monad Tree where
    return = Val
    (Val d) >>= h = h d
    (Choice l r) >>= h = Choice (l >>= h) (r >>= h)

(return and >>= are Haskell for the unit resp. the bind or Kleisli extension operation of a monad) and hence a functor

instance Functor Tree where
    fmap h ds = ds >>= return . h

(fmap is Haskell for the action of a functor on morphisms).

It is convenient to use the function flatten which calculates the list of leaves of a given tree:

flatten :: Tree u -> [ u ]
flatten (Val d) = [ d ]
flatten (Choice l r) = (flatten l) ++ (flatten r)

We implement enum^σ and questions^σ by mutual induction on σ ∈ Ty. The precise typings of the functions are enum ∈ (σ ∈ Ty) → Tree �σ� and questions ∈ (σ ∈ Ty) → [�σ� → �Bool�]. As usual, Haskell cannot express those subtleties due to its lack of dependent types, but we can declare

enum :: Sem u => Ty -> Tree u
questions :: Sem u => Ty -> [ u -> u ]

The base case for is straightforward: A boolean is true or false, and to know which one it is it suffices to know which one it is.

enum Bool = Choice (Val true) (Val false)

questions Bool = [ \ b -> b ]

The implementation of enum^{σ→τ} and questions^{σ→τ} proceeds from the idea that a function is determined by its graph: to know a function it suffices to know its value on all possible argument values. The main idea in the implementation of enum^{σ→τ} is therefore to start with enum^τ and to duplicate the tree for each question in questions^σ using the bind of Tree:

enum (sigma :-> tau) =
    fmap (lam sigma) (mkEnum (questions sigma) (enum tau))

mkEnum :: Sem u => [ u -> u ] -> Tree u -> Tree (u -> u)
mkEnum [] es = fmap (\ e -> \ d -> e) es
mkEnum (q : qs) es = (mkEnum qs es) >>= \ f1 ->
                     (mkEnum qs es) >>= \ f2 ->
                     return (\ d -> xif (q d) (f1 d) (f2 d))

questions^{σ→τ} produces the appropriate questions by enumerating σ and using questions from τ:

questions (sigma :-> tau) =
    [ \ f -> q (app f d) | d <- flatten (enum sigma),
                           q <- questions tau ]

As an example, the enumeration and questions for Bool → Bool return:

Choice
   (Choice
       (Val (lam Bool (\ d -> xif d true true)))
       (Val (lam Bool (\ d -> xif d true false))))
   (Choice
       (Val (lam Bool (\ d -> xif d false true )))
       (Val (lam Bool (\ d -> xif d false false))))

resp.

(\ f -> app f true :
   (\ f -> app f false :
       []))

We can look up an element in a tree by answering all the questions—this leads to find^σ ∈ [�Bool�] → Tree �σ� → �σ�, in Haskell:

find :: Sem u => [ u ] -> Tree u -> u
find [] (Val t) = t
find (a : as) (Choice l r) = xif a (find as l) (find as r)

We are now ready to implement quote^σ ∈ �σ�^{Set} → Tm σ, with Haskell typing

quote :: Ty -> SU -> UTm

by induction on σ ∈ Ty. As usual, the base case is easy:

quote Bool STrue = TTrue
quote Bool SFalse = TFalse

quote^{σ→τ} is more interesting: Our strategy is to map quote^τ ∘ f to the settheoretic enum^τ and to then build a tree of If expressions by using the syntactic questions^σ in conjunction with the syntactic find:

quote (sigma :-> tau) (SLam _ f) =
    lam sigma (\ t -> find [ q t | q <- questions sigma ]
                           (fmap (quote tau . f) (enum sigma)))

(Notice that in Haskell it is inferred automatically which semantics is meant where.)

As already discussed in the introduction, we implement normalization nf ∈ (σ ∈ Ty) → Tm σ → Tm σ by

nf :: Ty -> UTm -> UTm
nf sigma t = quote sigma (eval [] t)

Since we can infer types, we can implement nf^′ ∈ UTm → Maybe ((σ ∈ Ty) × Tm σ):

nf’ :: UTm -> Maybe (Ty, UTm)
nf’ t = do sigma <- infer [] t
           Just (sigma, nf sigma t)

We test our implementation with the example from the introduction:

b2b = Bool :-> Bool
once = Lam b2b "f" (Lam Bool "x" (App (Var "f") (Var "x")))
twice = Lam b2b "f" (Lam Bool "x" (App (Var "f")
                                     (App (Var "f") (Var "x"))))
thrice = Lam b2b "f" (Lam Bool "x" (App (Var "f")
                                     (App (Var "f")
                                       (App (Var "f") (Var "x")))))

and convince ourselves that nf^′ once == nf^′ thrice = true but nf^′ once == nf^′ twice = false. Since semantic equality is decidable we do not actually have to construct the normal forms to decide convertibility.

Since testing can only reveal the presence of errors we shall use the rest of this paper to prove that quote and hence nf behave correctly.

4 Correctness of quote

The main tool in our proof will be a notion of logical relations, a standard tool for characterization of definable elements in models of typed lambda calculi since the pioneering work of Plotkin [11].

Definition 4.1 (Logical Relations). We define a family of relations R^σ ⊆ Tm σ × �σ�^{Set} by induction on σ ∈ Ty as follows:

– if t =_{βη} True, then tR^{Bool}true; – if t =_{βη} False, then tR^{Bool}false;

– if for all u, d, uR^σd implies App t uR^τ f d, then tR^{σ→τ} f.

Logical relations obey the following Fundamental Theorem, a kind of substitution lemma for logical relations.

Lemma 4.2 (Fundamental Theorem of Logical Relations). If θR^Γ ρ and t ∈ Tm_Γ σ, then t[θ] R^σ �t�^{Set}_ρ . In particular, if t ∈ Tm σ, then tR^σ�t�^{Set}.

The main result required to see that quote is correct is the following lemma:

Lemma 4.3 (Main Lemma). If tR^σd, then t =_{βη} quote^σ d.

The proof of this lemma is the subject of the next section.

By correctness of quote we mean that it inverts set-theoretic evaluation of typed closed terms.

Theorem 4.4 (Main Theorem). If t ∈ Tm σ, then t =_{βη} quote^σ �t�^{Set}.

Proof. Immediate from the Fundamental Theorem and the Main Lemma. ⊓⊔

The (constructive) existence and correctness of quote has a number of straightforward important consequences.

Corollary 4.5 (Completeness). If t, t^′ ∈ Tm σ, then �t�^{Set} = �t^′�^{Set} implies t =_{βη} t^′.

Proof. Immediate from the Main Theorem. ⊓⊔

From soundness (Proposition 2.1) and completeness together we get that =_{βη} is decidable: checking whether t =_{βη} t^′ reduces to checking whether �t�^{Set} = �t^′�^{Set}, which is decidable as �−�^{Set} is computable and equality in finite sets is decidable.

Corollary 4.6. If t, t^′ ∈ Tm σ, then t =_{βη} t^′ iff quote^σ �t�^{Set} = quote^σ �t^′�^{Set}.

Proof. Immediate from soundness (Proposition 2.1) and the Main Theorem. ⊓⊔

This corollary shows that nf^σ = quote^σ ∘ �−�^{Set} : Tm σ → Tm σ indeed makes sense as normalization function: apart from just delivering, for any given typed closed term, some βη-equal term, it is actually guaranteed to deliver the same term for t, t^′, if t, t^′ are βη-equal (this is Church-Rosser for reduction-free normalization).

Note that although we only stated completeness and normalization for typed closed terms above, these trivially extend to all typed terms as opens terms can always be closed up by lambda-abstractions and this preserves βη-equality.

Corollary 4.7. If t, t^′ ∈ Tm σ and C t =_{βη} C t^′ for any C : Tm σ → Tm Bool, then t =_{βη} t^′. Or, contrapositively, and more concretely, if t, t^′ ∈ Tm (σ_1 → . . . → σ_n → Bool) and t ≠_{βη} t^′, then there exist u_1 ∈ Tm σ_1, . . . u_n ∈ Tm σ_n such that

nf^{Bool} (App (. . . (App t u_1) . . .) u_n) ≠ nf^{Bool} (App (. . . (App t^′ u_1) . . .) u_n)

Proof. This corollary does not follow from the statement of the Main Theorem, but it follows from its proof. ⊓⊔

Corollary 4.8 (Maximal consistency). If t, t^′ ∈ Tm σ and t ≠_{βη} t^′, then from the equation t = t^′ as an additional axiom one would derive True = False.

Proof. Immediate from the previous corollary. ⊓⊔

5 Proof of the main lemma

We now present the proof of the main lemma which was postponed in the previous section. To keep the proof readable, we write senum, squestions, sfind to emphasize the uses of the set-theoretic semantics instances of enum, questions, find, while the free semantics instances will be written as tenum, tquestions, tfind.

We first give the core of the proof and prove the lemmas this takes thereafter. Proof (of the Main Lemma). By induction on σ.

– Case Bool: Assume tR^{Bool}b. Then either t =_{βη} True and b = true, in which case we have

t =_{βη} True = quote^{Bool} true = quote^{Bool} b or t =_{βη} False and b = false, in which case we have

t =_{βη} False = quote^{Bool} false = quote^{Bool} b – Case σ → τ: Assume tR^{σ→τ} f, for all u, d, uR^σd implies App t uR^τ f d. We have

t =_{βη} Lam^σ x (App t (Var x)) =_{βη} (by Lemma 5.2 below)

Lam^σ x (App t (tfind [q (Var x) | q ← tquestions^σ ] tenum^σ ))

= Lam^σ x (tfind [q (Var x) | q ← tquestions^σ] (fmap (App t) tenum^σ)) =_{βη} (by Sublemma)

Lam^σ x (tfind [q (Var x) | q ← tquestions^σ] (fmap (quote^τ ∘ f) senum^σ))

= quote^{σ→τ} f The Sublemma is:

fmap (App t) tenum^σ (Tree =_{βη}) fmap (quote^τ ∘ f) senum^σ For proof, we notice that, by Lemma 5.1 (1) below for σ,

tenum^σ (Tree R^σ ) senum^σ Hence, by assumption

fmap (App t) tenum^σ (Tree R^τ ) fmap f senum^σ Hence, by IH of the Lemma for τ,

fmap (App t) tenum^σ (Tree =_{βη}) fmap (quote^τ ∘ f) senum^σ ⊓⊔ The proof above used two lemmas. One is essentially free, but the other is technical.

Lemma 5.1 (“Free” Lemma).

1. tenum^σ (Tree R^σ) senum^σ.

2. tquestions^σ [R^σ → R^{Bool}] squestions^σ.

Proof. Both follow from the Fundamental Theorem (Lemma 4.2), but in more detail, the proof is simultaneous for (1) and (2) by induction on σ.

– Case Bool: Trivial.

– Case σ → τ: Proof of (1) uses IH (2) for σ and IH (1) for τ; proof of (2) uses IH (1) for σ and IH (2) for τ. ⊓⊔ Lemma 5.2 (Technical Lemma). Define a relation < ⊆ UTm × [UTm → UTm] × Tree UTm by

t < (qs, ts) iff t =_{βη} tfind [q t | q ← qs] ts If t ∈ Tm σ, then t < (tquestions^σ , tenum^σ), i.e.,

t =_{βη} tfind [q t | q ← tquestions^σ ] tenum^σ

Proof. By induction on σ.

– Case Bool:

t =_{βη} If^′ t True False

= If^′ t (tfind [] (Val True)) (tfind [] (Val False)) = tfind [t] (Choice (Val True) (Val False)) = tfind [q t | q ← tquestions^{Bool}] tenum^{Bool}

– Case σ → τ:

t =_{βη} Lam^σx (App t (Var x)) =_{βη} (by Sublemma below, IH for σ)

Lam^σx (tfind [q (App t u) | u ← flatten tenum^σ , q ← tquestions^τ ]

(fmap (λg g (Var x)) (tmkenum tquestions^σ tenum^τ ))) = tfind [q (App t u) | u ← flatten tenum^σ , q ← tquestions^τ ] (fmap (λg Lam^σ x (g (Var x))) (tmkenum tquestions^σ tenum^τ ))) = tfind [q t | q ← tquestions^{σ→τ} ] tenum^{σ→τ}

The Sublemma is: If u ∈ UTm and u < (qs, us), then

App t u =_{βη} tfind [q (App t u^′) | u^′ ← flatten us, q ← tquestions^τ ]

(fmap (λg g u) (tmkenum qs tenum^τ ))

The proof is by “lockstep” induction on (qs, us).

• Case ([], Val u^⋆): Assume u =_{βη} tfind [] (Val u^⋆), i.e., u =_{βη} u^⋆. We get

App t u

=_{βη} (by assumption)

App t u^⋆

=_{βη} (by IH of the Lemma for τ)

tfind [q (App t u^⋆) | q ← tquestions^τ ] tenum^τ = tfind [q (App t u^⋆) | q ← tquestions^τ ]

(fmap (λg g u) (fmap (λvλu^′ v) tenum^τ )) = tfind [q (App t u^′) | u^′ ← [u^⋆], q ← tquestions^τ ] (fmap (λg g u) (tmkenum [] tenum^τ )) • Case (q : qs, Choice l r):

Assume u =_{βη} tfind (q u : [q^′u | q^′ ← qs]) (Choice l r), i.e., u =_{βη}

If^′ (q u) (tfind [q^′ u | q^′ ← qs] l) (tfind [q^′ u | q^′ ← qs] r). We get
     App t u

=_{βη} If^′ (q u) (App t u) (App t u) =_{βη} (by IH of the Sublemma for qs/l, qs/r)

If^′ (q u)

(tfind [q^′ (App t u^′) | u^′ ← flatten l, q^′ ← tquestions^τ ] (fmap (λg g u) (tmkenum qs tenum^τ ))) (tfind [q^′ (App t u^′) | u^′ ← flatten r, q^′ ← tquestions^τ ] (fmap (λg g u) (tmkenum qs tenum^τ ))) = tfind [] (Val (If^′ (q u)

(tfind [q^′ (App t u^′) | u^′ ← flatten l, q^′ ← tquestions^τ ] (fmap (λg g u) (tmkenum qs tenum^τ ))) (tfind [q^′ (App t u^′) | u^′ ← flatten r, q^′ ← tquestions^τ ] (fmap (λg g u) (tmkenum qs tenum^τ ))))) =_{βη} (by twice Lemma 5.3)

tfind ([q^′ (App t u^′) | u^′ ← flatten l, q^′ ← tquestions^τ ] ++ ([q^′ (App t u^′) | u^′ ← flatten r, q^′ ← tquestions^τ ] ++[])) ((fmap (λg g u) (tmkenum qs tenum^τ )) ≫= λv_0 (fmap (λg g u) (tmkenum qs tenum^τ )) ≫= λv_1 Val (If^′ (q u) v_0 v_1)) = tfind ([q^′ (App t u^′) | u^′ ← flatten l, q^′ ← tquestions^τ ] ++ [q^′ (App t u^′) | u^′ ← flatten r, q^′ ← tquestions^τ ]) (fmap (λg g u)

((tmkenum qs tenum^τ ) ≫= λg_0

(tmkenum qs tenum^τ ≫= λg_1 Val (λu^′ If^′ (q u^′) (g_0 u^′) (g_1 u^′))))

= tfind [q^′ (App t u^′) | t ← (flatten l) ++(flatten r), q^′ ← tquestions^τ ] (fmap (λg g u) (tmkenum (q : qs) tenum^τ )) = tfind [q^′ (App t u^′) | u^′ ← flatten (Choice l r), q^′ ← tquestions^τ ] (fmap (λg g u) (tmkenum (q : qs) tenum^τ )) ⊓⊔ In the last lemma, we used another lemma, a substitution lemma for tfind, which is easy to prove.

Lemma 5.3. Define a relation <> ⊆ [UTm]×Tree UTm by “lockstep” induction on the list and the tree by:

– if t ∈ UTm, then [] <> Val t;

– if as <> l, as <> r, then a : as <> Choice l r.

Then,

tfind (as ++bs) (ts ≫= h) =_{βη} tfind bs (h (tfind as ts))

Proof. By induction on (as, ts).

– Case ([], Val t):

tfind ([] ++bs) ((Val t) ≫= h) = tfind bs ((Val t) ≫= h) = tfind bs (h t) = tfind bs (h (tfind [] (Val t))

– Case (a : as, Choice l r):

tfind ((a : as) ++bs) ((Choice l r) ≫= h) = tfind (a : (as ++bs)) ((Choice l r) ≫= h) = If^′ a (tfind (as ++bs) (l ≫= h)) (tfind (as ++bs) (r ≫= h)) = (by IH for as/l, as/r)

If^′ a (tfind bs (h (tfind as l))) (tfind bs (h (tfind as r)))

=_{βη} tfind bs (h (If^′ a (tfind as l) (tfind as r))) = tfind bs (h (tfind (a : as) (Choice l r))) ⊓⊔ We note that, although not necessary for our purposes, the two last lemmas could have also been formulated generically for any semantics instead of just for the free one. To avoid introduction of extra notation, we refrained here from doing this.

6 Discussion and further work

Instead of decision trees we could have used a direct encoding of the graph of a function, we call this the truth-table semantics. However, this approach leads not only too much longer normal forms but also the semantic equality is less efficient. On the other hand it is possible to go further and use Binary Decision Diagrams (BDDs) [6] instead of decision trees. We plan to explore this in further work.

We have argued that λ^{→2} is the simplest λ-calculus with closed types, however we are confident that the technique described here works also for closed types in λ^{0+1×→} (the finitary λ calculus). We leave this extension for a journal version of this work.

One can go even further and implement finitary Type Theory, i.e. λ^{012ΣΠ} . This could provide an interesting base for a type-theoretic hardware description and verification language.

The approach presented here works only for calculi without typevariables. It remains open to see whether this approach can be merged with the the standard techniques for NBE for systems with type variables, leading to an alternative proof of completeness and maybe even finite completeness for the calculi discussed above.

References

1. T. Altenkirch, P. Dybjer, M. Hofmann, and P. Scott. Normalization by evaluation for typed lambda calculus with coproducts. In Proc. of 16th Annual IEEE Symposium on Logic in Computer Science, pages 303–310. IEEE CS Press, 2001.

2. T. Altenkirch, M. Hofmann, and T. Streicher. Categorical reconstruction of a reduction free normalization proof. In D. Pitt, D. E. Rydeheard, and P. Johnstone, editors, Category Theory and Computer Science, number 953 in LNCS, pages 182– 199, Springer-Verlag, 1995.

3. T. Altenkirch, M. Hofmann, and T. Streicher. Reduction-free normalisation for a polymorphic system. In Proc. of 11th Annual IEEE Symposium on Logic in Computer Science, pages 98–106. IEEE CS Press, 1996.

4. T. Altenkirch, M. Hofmann, and T. Streicher. Reduction-free normalisation for system F . Unpublished, available on WWW at

http://www.cs.nott.ac.uk/~txa/publ/f97.pdf, 1997.

5. U. Berger and H. Schwichtenberg. An inverse of the evaluation functional for typed λ-calculus. In Proc. of 6th Annual IEEE Symposium on Logic in Computer Science, pages 202–211. IEEE CS Press, 1991.

6. R. E. Bryant. Graph-based algorithms for boolean function manipulation. IEEE Trans. on Computers, 35(8):677–691, 1986.

7. D. J. Dougherty and R. Subrahmanyam. Equality between functionals in the presence of coproducts. Information and Computation, 157(1–2):52–83, 2000.

8. P. Dybjer and A. Filinski. Normalization and partial evaluation. In G. Barthe et al., editors, Applied Semantics, number 2395 in LNCS, pages 137–192. Springer- Verlag, 2002.

9. N. Ghani. Adjoint Rewriting. PhD thesis, LFCS, Univ. of Edinburgh, 1995.

10. N. Ghani. βη-equality for coproducts. In M. Dezani-Ciancaglini and G. D. Plotkin, editors, Typed Lambda Calculus and Applications, number 902 in LNCS, pages 171– 185. Springer-Verlag, 1995.

11. G. D. Plotkin. Lambda-definability and logical relations. Memorandum SAI-RM-4, Lab. for Artif. Intell., Univ. of Edinburgh, Oct. 1973.